

It's halftime in the semi-final against Czechoslovakia and you're 2-0 down.

As the team troop back into the changing rooms you wonder whether you should change your tactics. The 4-2-4 system you elected to play seems to be giving the Czechs a lot of possession in midfield and your sweeper has been caught out twice by their tendency to hit early crosses to their centre forward Nehoda — who scored both the goals.

You seemed to have the easier task too, for in the first semi-final West Germany and Brazil — two of the teams you had fancied for the title — had battled to a 2-2 draw, with Brazil romping through 5-2 in extra time.

Luckily you had the foresight to get your scouts to compile a thorough dossier on Brazil and there ought to be no surprises when you meet them. Perhaps you'd have been better off sending a scout to watch the Czechs.

Well you'd better pull another player back into midfield and also try to close down the Czechs' space in the centre — Panenka has been controlling the midfield.

And what about a substitution? Several of your players have been giving away possession.

It's not as though you haven't had chances, Morley was closest rounding two defenders before he shot wide, if that had gone in...



RUNS ON A SHARP MZ-80K IN 26K

BY TERRY ALLEN

WORLD CUP

World Cup Manager sets out to give you all the frustrations of sitting on the bench when the national side you manage go a goal down 20 minutes from time in a crucial match.

And hopefully, with some careful attention to tactics and good substitutions in critical games, it should give you all the elation of guiding your country's team through to the cham-

pionships of the 1982 World Cup Finals.

The game runs on a Sharp MZ-80K in a massive 26K but is packed with REM statements and there is not a PEEK or a POKE in sight so transferring it to another machine should not cause too many problems.

It will let you manage any of the 24 countries involved in the 1982 World Cup but expect to have a more difficult time if you pick El Salvador than if you choose Brazil.

It invites you to enter your team and then publishes the draw for the competition with initially six groups of four. Set out the countries you think will be stumbling blocks to your continuation in the competition and send off your scouts to watch them.

Remember you can get through by finishing second in your group so it may not be necessary to watch all three teams in your group. The first game comes up and a dossier on the opposition is printed out. Pay careful attention because you will have to find good tactical answers to the opposition's plays.

There are other considerations too, choose your formation carefully and if you need to boost your goal difference to qualify, try to manufacture an open game.

Turn to page 35

```
4 DIM Q(4),W(4),D9(4),F9(4),A9(4),G2(4,3),D8(4),T9(4),P9(4),R(24)
5 DIM S8(4),N1(2),G(24),M(6,4),T1(3,3),T2(3,3),T3(3,3),T4(3,3),T4$(20)
6 V=0:B4=0
7 DEF FNA(X)=INT(RND(1)*X)+1
10 DIM A$(24):DIM F(24):DIM H1(3,3)
12 DIM A1(3,3):DIM M1(24,21):DIM M2(24)
14 DIM T5(24),T6(24),T7(24),T8(24)
16 DIM S5(24),S6(24),S7(24),Z9$(12)
18 DIM A1$(9),A2$(9),U2(11)
20 DATA BRAZIL, ARGENTINA, W.GERMANY, YUGOSLAVIA
30 DATA HUNGARY, BELGIUM, ITALY, POLAND, FRANCE
40 DATA CZECHOSLOVAKIA, SPAIN, ENGLAND, SCOTLAND, N. IRELAND, PERU
50 DATA AUSTRIA, CHILE, "EL SALVADOR", HONDURAS
60 DATA ALGERIA, N.ZEALAND, CAMEROON, KUWAIT, RUSSIA
62 DATA 1,1,0,1,2,1,1,0,0,0,0,1,1,2,1,0,2,2
64 DATA " HITS THE BALL TOO FAR AHEAD OF ", "S PASS IS MISCONTROLLED BY "
66 DATA " LOSES THE BALL IN A MISUNDERSTANDING WITH ", " HITS A LONG BALL TO "
68 DATA " PASSES WIDE TO "
69 DATA " CROSSES TO ", " PLAYS A HIGH BALL TO ", " COLLECTS A PASS FROM "
70 DATA " PLAYS A ONE-TWO WITH "
72 DATA " IS SENT AWAY BY ", " WINS A FREE KICK FROM ", " MOVES INSIDE "
74 DATA " SWERVES PAST ", " DUMMIES ", " RUNS ROUND ", " RACES PAST "
76 DATA " SHUGGS OFF A CHALLENGE FROM "
82 DATA " IS TACKLED BY ", "S PASS IS CUT OUT BY "
84 DATA "S CROSS IS HEADED AWAY BY "
90 DATA 5,15,30,30,5,15,15,30,5,5,30,15,15,5,30,30,15,5
92 DATA 5,30,15,15,5,30,30,15,5
94 DATA "S CHIP FOR GOAL IS HEADED AWAY BY ", " LOSES THE BALL TO "
96 DATA " IS TACKLED BY ", "S VOLLEY IS PUNCHED CLEAR BY "
98 DATA "S FIERCE DRIVE IS WELL SAVED BY "
99 DATA " BLASTS THE BALL WIDE ", " ROUNDS THE "KEEPER AND MISKICKS"
99 DATA " BEATS TWO DEFENDERS AND SHOOTS WIDE", " STUMBLES AND LOSES THE BALL"
100 DATA "S CHIP FOR GOAL FINDS THE BACK OF THE NET", " BLASTS THE BALL HOME"
102 DATA " VOLLEYS INTO THE ROOF OF THE NET", " TAPS THE BALL IN FROM 5 YARDS"
104 DATA " HITS A LOW SHOT INTO THE NET", "S SHOT BOUNCES UNDER THE KEEPER"
106 DATA "S SOFT SHOT IS DROPPED BY THE KEEPER", " LETS FLY WITH A LONG SHOT"
108 DATA " STEPS INSIDE THE FULLBACK, ROUNDS THE KEEPER AND SLOTS HOME"
```

```

188 GOSUB 11062
190 GOSUB 11070
194 GOSUB 11094
196 GOSUB 11110
200 GOSUB 12100
210 GOSUB 5400
220 GOSUB 12300
230 GOSUB 12615
270 PRINT "E"
280 PRINT "PLEASE CHOOSE YOUR TEAM"
290 PRINT "ENTER A NUMBER FROM THIS LIST"
292 N=1
295 GOSUB 1295
330 GOSUB 1330
332 GOSUB 670
400 PRINT "E"
410 PRINT "YOU ARE MANAGER OF ":A$(T)
420 PRINT "YOUR AIM IS TO WIN THE 1982 WORLD CUP"
440 GOSUB 9930
480 FOR I=1 TO 6
490 S=0
500 Y=FNA(24)
510 IF G(V)<>0 THEN GOTO 500
515 S=S+1
518 G(V)=I
520 M(I,S)=Y
530 IF S=4 THEN GOTO 560
540 GOTO 500
560 NEXT I
562 PRINT "E": "THE DRAW IS AS FOLLOWS"
564 PRINT " "
570 FOR I=1 TO 3
575 PRINT " "
580 PRINT "GROUP ":STR$(I);TAB(20);"GROUP ":STR$(I+3)
582 PRINT " "
590 FOR J=1 TO 4
600 PRINT A$(M(I,J));TAB(20);A$(M(I+3,J))
610 NEXT J
620 NEXT I
625 PRINT " "
628 FOR I=1 TO 5000:NEXT I

```

```

630 GOSUB 9900
635 GOSUB 6000
652 GOSUB 5000
654 GOSUB 5100
656 GOSUB 9900
658 GOSUB 5200
660 GOSUB 4200
661 GOSUB 3000
662 GOSUB 3200
663 GOSUB 9900
664 IF E2>E1 THEN B2=X2:B3=X1
665 IF E1>E2 THEN B2=X1:B3=X2
666 PRINT "E":PRINT "CONGRATULATIONS TO ":A$(B2)
667 PRINT "1982 WORLD CUP WINNERS"
668 END
669 PRINT "E"
670 PRINT "NOW YOU MUST NAME YOUR TEAM "
680 DIM K$(12)
690 PRINT "KEY IN THE PLAYER'S NAMES AS PROMPTED"
700 D$="GOALKEEPER ":I=1
705 GOSUB 14000
710 FOR I=2 TO 5
720 D$="DEFENDER "+STR$(I-1)
730 GOSUB 14000
740 NEXT I
750 FOR I=6 TO 8
760 D$="MIDFIELD PLAYER "+STR$(I-5)
770 GOSUB 14000
780 NEXT I
790 FOR I=9 TO 11
800 D$="STRIKER "+STR$(I-8)
810 GOSUB 14000
820 NEXT I
822 D$="SUBSTITUTE ":I=12
824 GOSUB 14000
826 RETURN
870 END
1295 PRINT " "
1300 FOR I=1 TO 12
1302 IF X=1 THEN GOTO 1310
1304 IF (T-I)*(T-12-I)<>0 THEN GOTO 1310
1305 IF T=I+12 THEN GOTO 1308

```

MANAGER

```

1306 PRINT TAB(20);STR$(I+12);".":A$(I+12)
1307 GOTO 1320
1308 PRINT STR$(I);".":A$(I)
1309 GOTO 1320
1310 PRINT STR$(I);".":A$(I);TAB(20);STR$(I+12);".":A$(I+12)
1320 NEXT I
1322 PRINT " "
1324 PRINT " "
1325 RETURN
1330 INPUT "WHICH TEAM?":T$
1335 T=VAL(T$)
1340 IF T<1 THEN GOTO 1350
1342 IF T>24 THEN GOTO 1350
1344 GOTO 1370
1350 PRINT "PLEASE TYPE A NUMBER BETWEEN 1 AND 24"
1360 GOTO 1330
1370 RETURN
3000 PRINT "E"
3010 PRINT "BARCELONA 8th JULY 1982"
3020 PRINT " "
3030 PRINT "WORLD CUP SEMI-FINAL"
3040 PRINT " "
3042 PRINT A$(S8(1));" U. ":A$(S8(3))
3043 X3=1
3044 GOSUB 9930
3045 X1=S8(1):X2=S8(3)
3046 IF X1=T THEN Z=X2:GOSUB 9000:E1=B:E2=C:GOSUB 4475:GOTO 3052
3048 IF X2=T THEN Z=X1:GOSUB 9000:E1=C:E2=B:GOSUB 4475:GOTO 3052
3050 GOSUB 4400
3052 M2(X1)=M2(X1)+1
3053 M1(X1,6)=X2:M1(X1,13)=E1:M1(X1,20)=E2
3054 M2(X2)=M2(X2)+1
3055 M1(X2,6)=X1:M1(X2,13)=E2:M1(X2,20)=E1
3056 IF (X1-T)*(X2-T)<>0 THEN GOSUB 9900
3060 PRINT "E"
3070 PRINT "SEVILLE 8th JULY 1982"
3080 PRINT " "
3090 PRINT "WORLD CUP SEMI-FINAL"
3100 PRINT " "
3102 E2=0:E1=0
3104 PRINT A$(S8(2));" U. ":A$(S8(4))
3106 GOSUB 9930
3107 X1=S8(2):X2=S8(4)
3108 IF X1=T THEN Z=X2:GOSUB 9000:E1=B:E2=C:GOSUB 4475:GOTO 3112
3109 IF X2=T THEN Z=X1:GOSUB 9000:E1=C:E2=B:GOSUB 4475:GOTO 3112
3111 GOSUB 4400
3112 M2(X1)=M2(X1)+1
3114 M2(X2)=M2(X2)+1
3116 M1(X1,6)=X2:M1(X1,13)=E1:M1(X1,20)=E2
3118 M1(X2,6)=X1:M1(X2,13)=E2:M1(X2,20)=E1
3120 IF (X1-T)*(X2-T)<>0 THEN GOSUB 9900
3130 RETURN

```



```

3200 U=1
3202 PRINT "E"
3205 PRINT "MADRID 11th JULY 1982"
3300 FOR I=1 TO 6
3310 PRINT " "
3320 NEXT I
3330 PRINT "1982 WORLD CUP FINAL"
3340 PRINT "===== "
3342 PRINT A$(N1(1)); " U. "; A$(N1(2))
3350 GOSUB 9930
3360 GOSUB 9930
3362 PRINT "E"
3370 E1=0:E2=0
3380 X1=N1(1):X2=N1(2)
3382 IF X1=T THEN Z=X2:GOSUB 9000:E1=B:E2=C:GOTO 3400
3384 IF X2=T THEN Z=X1:GOSUB 9000:E1=C:E2=B:GOTO 3400
3390 GOSUB 4400
3400 RETURN
3410 REM CHECK IF YOU'RE PLAYING
3420 FOR I=1 TO W6
3430 IF (M(I,H2)-T)*(M(I,A2)-T)=0 THEN GOSUB 3500
3440 NEXT I
3442 RETURN
3450 REM GET SCOUT'S REPORT
3452 D4=0
3455 IF (Z-S1)*(Z-S2)*(Z-S3)*(Z-S4)=0 THEN D4=1
3460 IF D4=1 THEN J$="YOUR SCOUTS HAVE BEEN WATCHING "+A$(Z):GOSUB 15000
3462 E=Z
3465 IF D4=1 THEN GOSUB 9900
3467 GOSUB 8400
3470 RETURN
3500 REM SLOT IN VALUE OF OPPOSITION
3510 IF M(I,H2)<>T THEN Z=M(I,H2)
3520 IF M(I,H2)=T THEN Z=M(I,A2)
3530 GOSUB 9000
3540 RETURN
3800 REM TYPE IN OPPOSITION TEAM
3810 PRINT "E":PRINT "DO YOU WISH TO TYPE IN THE OPPOSITION  PLAYERS' NAMES?"
3820 INPUT "TYPE Y OR N ";D3$
3822 IF ASC(LEFT$(D3$,1))<>89 THEN GOSUB 12570: RETURN
3823 FOR I=1 TO 12
3825 PRINT "TYPE IN THE NAME OF PLAYER":I
3830 INPUT Z$(I)
3832 NEXT I
3840 RETURN
4000 REM SEMI-FINALISTS
4010 S8(I)=T9(J)
4020 RETURN
4200 PRINT "E"
4210 PRINT "1982 WORLD CUP SEMI-FINALS"
4220 PRINT "===== "
4230 FOR I=1 TO 4
4240 PRINT " "
4250 NEXT I
4260 PRINT A$(S8(1)):TAB(16); "U.          "; A$(S8(3))
4270 FOR I=1 TO 4
4280 PRINT " "
4290 NEXT I
4300 PRINT A$(S8(2)):TAB(16); "U.          "; A$(S8(4))
4302 FOR I=1 TO 6
4304 PRINT " "
4306 NEXT I
4310 GOSUB 9900
4320 RETURN
4400 REM PLAY SEMI-FINAL, FINAL
4410 FOR I=1 TO 90
4420 IF RND(1)>86/90 THEN GOSUB 4600
4425 IF I=45 THEN GOSUB 4900
4430 IF I/10=INT(I/10) THEN GOSUB 4800
4440 NEXT I
4450 IF E1=E2 THEN GOSUB 4700
4460 PRINT "FINAL SCORE "
4470 PRINT A$(X1):E1; " "; A$(X2):E2
4475 W5=W5+1
4476 IF W5=1 THEN RETURN
4480 IF E1>E2 THEN N1(W5)=X1
4490 IF E2>E1 THEN N1(W5)=X2
4510 RETURN
4600 R7=RND(1)
4601 PRINT "E"
4602 R9=R(X1)/(R(X1)+R(X2))
4605 IF R7>R9 THEN GOTO 4600
4610 IF R7<R9 THEN J$=A$(X1)+" HAVE SCORED AFTER "+STR$(I)+" MINUTES"
4611 IF I=1 THEN J$=LEFT$(J$,LEN(J$)-1)
4612 IF R7<R9 THEN GOSUB 15000
4615 IF R7<R9 THEN E1=E1+1
4620 IF R7>R9 THEN J$=A$(X2)+" HAVE SCORED AFTER "+STR$(I)+" MINUTES"
4621 IF R7>R9 THEN GOSUB 15000
4625 IF R7>R9 THEN E2=E2+1
4627 GOSUB 9930
4630 RETURN
4700 PRINT "E"
4702 PRINT "AFTER EXTRA TIME"
4705 R4=FNA(2)
4710 IF R4=1 THEN E1=E1+FNA(3)
4720 IF R4=2 THEN E2=E2+FNA(3)
4725 GOSUB 9930
4730 RETURN
4800 PRINT "E"
4805 PRINT "SCORE AFTER ":I;" MINUTES ":";"
4810 PRINT A$(X1); " "; E1; " "; A$(X2); " "; E2
4815 GOSUB 9930
4820 RETURN
4900 PRINT " "
4910 PRINT "HALF TIME SCORE:"

```

Continued from page 33

Then it's out of your hands and the two teams are up against one another with a running commentary of events. Keep a close check on which of your players seem to lose possession and play badly, these are the one's you will want to substitute. A good substitution will improve your play — a bad one will give the opposition more chance.

Also watch for which team is gaining the most possession in midfield — you may need to change your formation or tactics here. Check on how the goals are scored — it could be important to tighten up your defence to certain situations or to find out which of your own attacking plays is paying dividends.

Half-time comes around and this is your chance to change your tactics if you have noticed things going wrong.

At the end of the day, win, lose or draw, you will have to sit back and watch the other results of the day being typed out and see if your predictions are coming true.

And on you go, hopefully towards a place in the final.

The game is run on exactly the same lines as the World Cup with the winners of Group 1 going into Group A and the runners-up into Group C; Group 2 winners into Group B and runners-up into Group D; Group 3 winners into Group A and runners-up into Group C; Group 4 winners into Group C and runners-up into Group A; Group 5 winners into Group D and runners-up into Group B; Group 6 winners into Group B and runners-up into Group D.

The first semi-final is between the winners of Group A and C and the other between the winners of Group B and D.



```

4920 PRINT " "
4930 PRINT A$(X1): " "E1: " "A$(X2): " "E2
4940 PRINT " "
4945 GOSUB 9930
4950 RETURN
5000 REM INITIAL GROUP FIXTURES
5004 W6=6
5005 D1=12
5010 H2=1:A2=2:GOSUB 5020
5011 H2=3:A2=4:GOSUB 5020
5012 H2=1:A2=3:GOSUB 5020
5013 H2=2:A2=4:GOSUB 5020
5014 H2=1:A2=4:GOSUB 5020
5015 H2=2:A2=3:GOSUB 5020
5016 RETURN
5020 D1=D1+1
5021 GOSUB 3410
5022 PRINT " "
5024 PRINT "RESULTS FOR "D1: " JUNE '82"
5026 PRINT " "
5027 FOR I=1 TO 6
5028 H=M(I,H2):A=M(I,A2)
5029 IF H=T THEN B=P2:C=P3:GOTO 5035
5030 IF T=A THEN B=P3:C=P2:GOTO 5035
5032 GOSUB 7000
5035 PRINT "GROUP":I: " "A$(H):B: " "A$(A):C
5040 NEXT I
5041 IF D1=18 THEN GOSUB 7200
5042 PRINT " "
5043 IF D1=18 THEN RETURN
5044 PRINT "PRESS L TO DISPLAY LEAGUE TABLES"
5045 PRINT " " ANY OTHER KEY TO CONTINUE"
5046 GET Z$
5048 IF Z$="" THEN GOTO 5046
5049 IF Z$="L" THEN GOSUB 7200
5050 RETURN
5100 REM PRINT GROUPS A-D
5105 PRINT " "
5106 PRINT "GROUPS FOR THE SECOND STAGE ARE :-"
5107 PRINT " "
5110 FOR I=1 TO 2
5120 PRINT "GROUP ":CHR$(64+I):TAB(20):"GROUP ":CHR$(66+I)
5130 PRINT " "
5140 FOR J=1 TO 3
5150 PRINT A$(G2(I,J)):TAB(20):A$(G2(I+2,J))
5160 NEXT J
5165 PRINT " "
5170 NEXT I
5189 RETURN
5200 REM SECOND STAGE FIXTURES
5205 W6=4
5210 D1=27
5220 H2=1:A2=2:GOSUB 5300
5230 H2=3:A2=1:GOSUB 5300
5240 H2=2:A2=3:GOSUB 5300
5250 RETURN
5300 REM SECOND STAGE FIXTURES
5302 GOSUB 8310
5310 GOSUB 3410
5322 PRINT " "
5324 PRINT "RESULTS FOR "D1: " JUNE '82"
5326 PRINT " "
5327 FOR I=1 TO 4
5329 H=G2(I,H2):A=G2(I,A2)
5330 IF H=T THEN B=P2:C=P3:GOTO 5335
5331 IF T=A THEN B=P3:C=P2:GOTO 5335
5332 GOSUB 7000
5335 PRINT "GROUP ":CHR$(I+64): " "A$(H):B: " "A$(A):C
5340 NEXT I
5341 IF D1=29 THEN GOSUB 7200
5342 PRINT " "
5343 IF D1=29 THEN RETURN
5344 PRINT "PRESS L TO DISPLAY LEAGUE TABLES"
5345 PRINT " " ANY OTHER KEY TO CONTINUE"
5346 GET Z$
5348 IF Z$="" THEN GOTO 5346
5349 IF Z$="L" THEN GOSUB 7200
5350 D1=D1+1
5352 RETURN
5400 REM GENERATE TEAM TACTICS
5410 FOR I=1 TO 24
5412 S5(I)=FNA(3)
5414 S6(I)=FNA(3)
5416 S7(I)=FNA(3)
5420 T5(I)=FNA(3)
5430 T6(I)=FNA(3)
5440 T7(I)=FNA(3)
5450 T8(I)=FNA(3)
5460 NEXT I
5470 RETURN
5480 REM GET DESCRIPTION OF FREEKICK TACTICS
5490 IF F1=1 THEN C$=" DIRECT SHOT "
5500 IF F1=2 THEN C$=" CHIP TO THE FAR POST "
5510 IF F1=3 THEN C$=" TEE-UP FOR SHOT "
5520 RETURN
5530 REM DESCRIPTION OF CORNER TACTICS
5540 IF F1=1 THEN C$=" PLAY IT SHORT "
5550 IF F1=2 THEN C$=" AIM FOR THE FAR POST "
5560 IF F1=3 THEN C$=" AIM FOR THE NEAR POST "
5570 RETURN
5580 REM DESCRIPTION OF ATTACK TACTICS
5590 IF F1=1 THEN C$=" FAST BREAK "
5600 IF F1=2 THEN C$=" EARLY CROSS "
5610 IF F1=3 THEN C$=" SLOW BUILD-UP "
5620 RETURN
5630 REM MIDFIELD TACTICS
5640 IF F1=1 THEN C$=" TACKLE HARD "
5650 IF F1=2 THEN C$=" MARK TIGHT "
5660 IF F1=3 THEN C$=" FIND SPACE "
5670 RETURN
5680 REM FREEKICK DEFENCE
5690 IF F1=1 THEN C$=" MAKE A WALL "
5700 IF F1=2 THEN C$=" MARK MAN-TO-MAN "
5710 IF F1=3 THEN C$=" RUSH THE KICKER "
5720 RETURN
5730 REM CORNER DEFENCE
5740 IF F1=1 THEN C$=" MOVE OUT FAST "
5750 IF F1=2 THEN C$=" RELY ON THE GOALKEEPER "
5760 IF F1=3 THEN C$=" USE CLOSE MARKING "
5770 RETURN
5780 REM DEFENCE TACTICS
5790 IF F1=1 THEN C$=" USE A SWEEPER "
5800 IF F1=2 THEN C$=" INTERCEPT PASSES "
5810 IF F1=3 THEN C$=" SHADOW ATTACKERS "
5820 RETURN
5830 IF F1=1 THEN C$=" A 4-3-3 "
5840 IF F1=2 THEN C$=" A 4-4-2 "
5850 IF F1=3 THEN C$=" A 4-2-4 "
5860 RETURN
6000 REM CHOOSE TEAMS FOR SCOUTS TO WATCH
6005 PRINT " "
6010 PRINT "SCOUTS CAN COMPILE DOSSIERS ON 4 TEAMS"
6020 PRINT "ENTER FOUR NUMBERS FROM THE FOLLOWING:"
6030 PRINT "SEPARATED BY COMMAS"
6040 X=2
6050 GOSUB 1295
6060 INPUT S1,S2,S3,S4
6070 IF (S1-T)*(S2-T)*(S3-T)*(S4-T)=0 THEN GOTO 6200
6072 IF (S1-S2)*(S1-S3)*(S1-S4)*(S2-S3)*(S2-S4)*
(S3-S4)=0 THEN 6200
6080 IF S1<1 THEN GOTO 6200
6090 IF S2<1 THEN GOTO 6200
6100 IF S3<1 THEN GOTO 6200
6110 IF S4<1 THEN GOTO 6200
6120 IF S4>24 THEN GOTO 6200
6130 IF S3>24 THEN GOTO 6200
6140 IF S2>24 THEN GOTO 6200
6150 IF S1>24 THEN GOTO 6200
6160 RETURN
6200 PRINT "RE-ENTER 4 NUMBERS FROM ABOVE LIST"
6210 GOTO 6060
7000 REM PLAYS TEAM H U. TEAM A
7010 REM MATCH SCORE IS B C
7011 B=0:C=0
7012 IF FNA(100)<60 THEN GOTO 7030
7020 B=H1(F(H),F(A)):C=A1(F(H),F(A))
7030 B1=FNA(8):C1=FNA(8)
7032 B1=B1+R(H):C1=C1+R(A)
7040 D=INT((ABS(B1-C1))/5)
7042 IF D<>0 THEN 7080
7050 B=B+FNA(2)-1
7060 C=C+FNA(2)-1
7070 IF FNA(2)=2 THEN GOTO 7100
7072 IF R(H)>R(A) THEN B=B+FNA(2)-1
7074 IF R(A)>R(H) THEN C=C+FNA(2)-1
7078 GOTO 7100
7080 IF B1>C1 THEN B=B+D
7090 IF C1>B1 THEN C=C+D
7092 IF FNA(500)>498 THEN B=B+FNA(3)
7094 IF FNA(500)>498 THEN C=C+FNA(3)
7100 M2(H)=M2(H)+1:M2(A)=M2(A)+1
7110 M1 (H,M2(H))=A
7120 M1 (A,M2(A))=H
7130 M1 (H,M2(H)+7)=B
7140 M1 (A,M2(A)+7)=C
7150 M1 (H,M2(H)+14)=C
7160 M1 (A,M2(A)+14)=B
7170 RETURN
7200 REM
7201 PRINT "CALCULATING LEAGUE TABLES"
7202 IF W6=4 THEN GOSUB 8300
7204 IF W6=4 THEN GOTO 7210
7206 K1=1:K2=3:J1=4
7210 FOR I=1 TO W6
7220 GOSUB 7500
7230 GOSUB 7900
7235 IF W6+I-9=0 THEN PRINT "CALCULATING LEAGUE TABLES"
7240 NEXT I
7250 RETURN
7500 REM WORKOUT LEAGUE POSITIONS
7502 FOR J=1 TO J1
7504 Q(J)=0:F9(J)=0:A9(J)=0:P9(J)=0:W(J)=0:D8(J)=0:D9(J)=0
7506 NEXT J
7510 FOR J=1 TO J1
7520 T9(J)=M(I,J)
7530 FOR K=K1 TO K2
7540 IF M1(T9(J),K)=0 THEN GOTO 7600
7550 F9(J)=M1(T9(J),K+7)+F9(J)
7560 A9(J)=M1(T9(J),K+14)+A9(J)
7562 Z7=M1(T9(J),K+7)
7564 Z6=M1(T9(J),K+14)
7570 D8(J)=Z7-Z6+D8(J)
7580 Q(J)=Q(J)+1
7585 IF Z7>Z6 THEN D9(J)=D9(J)+1
7590 IF Z6>Z7 THEN W(J)=W(J)+1
7592 IF Z6<Z7 THEN P9(J)=P9(J)+2
7594 IF Z6=Z7 THEN P9(J)=P9(J)+1
7600 NEXT K
7605 NEXT J
7606 F2=0

```

```

7610 FOR J=1 TO J1-1
7620 IF P9(J)<P9(J+1) THEN GOSUB 7800
7630 IF P9(J)=P9(J+1) THEN GOSUB 7700
7640 NEXT J
7650 IF F2=1 THEN GOTO 7606
7660 RETURN
7700 REM TEST GOAL DIFFERENCE
7710 IF D8(J)<D8(J+1) THEN GOSUB 7800
7720 IF D8(J)>D8(J+1) THEN RETURN
7730 IF F9(J)<F9(J+1) THEN GOSUB 7800
7740 IF F9(J)>F9(J+1) THEN RETURN
7750 IF FNA(2)=1 THEN GOSUB 7800
7760 RETURN
7800 REM SWAP TEAMS POSITIONS
7810 F3=T9(J):T9(J)=T9(J+1):T9(J+1)=F3
7820 F3=F9(J):F9(J)=F9(J+1):F9(J+1)=F3
7825 F3=A9(J):A9(J)=A9(J+1):A9(J+1)=F3
7830 F3=D8(J):D8(J)=D8(J+1):D8(J+1)=F3
7835 F3=D9(J):D9(J)=D9(J+1):D9(J+1)=F3
7840 F3=P9(J):P9(J)=P9(J+1):P9(J+1)=F3
7845 F3=W(J):W(J)=W(J+1):W(J+1)=F3
7850 F3=Q(J):Q(J)=Q(J+1):Q(J+1)=F3
7860 F2=1
7870 RETURN
7900 REM PRINT LEAGUE POSITIONS
7902 PRINT " "
7904 F$="F ":A4$="A "
7905 IF W6=6 THEN B$=CHR$(I+48)
7906 IF W6=4 THEN B$=CHR$(I+64)
7910 PRINT "GROUP "B$:" LEAGUE TABLE"
7912 FOR J=1 TO J1
7914 IF F9(J)>9 THEN F$="F "
7916 IF A9(J)>9 THEN A4$="A "
7918 NEXT J
7920 PRINT " "
7925 PRINT TAB(20):" P W D L ":F$:A4$:"Pts."
7930 FOR J=1 TO J1
7932 P4$="":P5$=""
7935 L9=Q(J)-W(J)-D9(J)
7936 IF F$="F " THEN GOTO 7990
7937 IF A4$="A " THEN GOTO 7994
7940 V1$=STR$(J)+". "+A$(T9(J))
7942 PRINT V1$:TAB(20):Q(J):W(J):D9(J):L9:P4$:F9(J):P5$:
A9(J):" ":P9(J)
7943 IF W6+J-5=0 THEN GOSUB 4000
7944 IF W6=4 THEN 7950
7946 IF J<3 THEN GOSUB 8900
7950 NEXT J
7951 IF W6=6 THEN 7956
7952 IF I<4 THEN 7985
7954 GOTO 7960
7956 IF I<3 THEN 7965
7958 IF (I-4)*(I-5)=0 THEN 7985
7960 PRINT "PRESS ANY KEY TO CONTINUE"
7970 GET Z$
7980 IF Z$="" THEN GOTO 7970
7985 RETURN
7990 IF F9(J)<10 THEN P4$=" "
7992 GOTO 7937
7994 IF A9(J)<10 THEN P5$=" "
7996 GOTO 7940
8300 K1=4:K2=5:J1=3
8310 FOR I=1 TO 4
8320 FOR J=1 TO 3
8330 M(I,J)=G2(I,J)
8340 NEXT J
8350 NEXT I
8360 RETURN
8400 REM SCOUT'S REPORT
8412 PRINT"PROFILE ON ":A$(E):PRINT " "
8414 GOSUB 8600
8415 PRINT " "
8418 F1=S5(E)
8420 GOSUB 5780
8425 PRINT " "
8430 PRINT "IN DEFENCE THEY":C$
8432 PRINT " "
8440 F1=S6(E):GOSUB 5580
8450 PRINT "IN ATTACK THEY USE THE":C$
8452 PRINT " "
8460 F1=S7(E):GOSUB 5630
8470 PRINT "IN MIDFIELD THEY":C$
8472 PRINT " "
8473 IF D4=0 THEN GOSUB 9900:RETURN
8480 F1=T5(E):GOSUB 5480
8492 PRINT "IN FREEKICKS AROUND THE BOX THEY USE A":PRINTC$
8490 F1=T6(E):GOSUB 5680
8500 PRINT "AND DEFENDING AGAINST FREEKICKS THEY":PRINTC$
:PRINT " "
8510 F1=T7(E):GOSUB 5530
8520 PRINT"FROM A CORNER THEY":C$
8530 F1=T8(E):GOSUB 5730
8540 PRINT"WHEN FACED WITH A CORNER THEY":PRINTC$:PRINT " "
8550 F1=F(E):GOSUB 5830
8560 PRINT"THEY NORMALLY PLAY":C$:"FORMATION"
8562 PRINT " "
8565 GOSUB 9900
8570 RETURN
8600 REM PRINT PAST RESULTS
8605 IF M2(E)=0 THEN RETURN
8610 PRINT " "
8620 PRINT"RESULTS SO FAR IN WORLD CUP"
8622 PRINT " "
8630 FOR I=1 TO M2(E)
8640 PRINTA$(E):M1(E,I+7):" ":A$(M1(E,I)):M1(E,I+14)

```

```

8650 NEXT I
8652 PRINT " "
8654 GOSUB 9900
8660 RETURN
8670 REM SELECT TACTICS
8672 PRINT"0"
8675 PRINT"WHAT FORMATION SHALL WE PLAY BOSS?"
8680 PRINT " "
8685 FOR F1 =1 TO 3
8690 GOSUB 5830
8695 PRINTF1:" ":C$
8700 NEXT F1
8702 PRINT " "
8705 GOSUB 9960
8710 F(T)=E9
8712 PRINT " "
8715 PRINT"WHAT ARE THE DEFENSIVE PLANS BOSS?"
8720 PRINT " "
8725 FOR F1 =1 TO 3
8730 GOSUB 5780
8735 PRINTF1:" ":C$
8740 NEXT F1
8742 PRINT " "
8745 GOSUB 9960
8747 PRINT " "
8750 S5(T)=E9
8755 PRINT"WHAT ARE THE ATTACKING PLANS GUV?"
8760 PRINT " "
8765 FOR F1 =1 TO 3
8770 GOSUB 5580
8775 PRINTF1:" ":C$
8780 NEXT F1
8782 PRINT " "
8785 GOSUB 9960
8790 PRINT " "
8795 S6(T)=E9
8800 PRINT"WHAT IS THE MIDFIELD SCHEME BOSS?"
8802 PRINT " "
8804 FOR F1 =1 TO 3
8806 GOSUB 5630
8808 PRINTF1:" ":C$
8810 NEXT F1
8812 PRINT " "
8814 GOSUB 9960
8816 PRINT " "
8818 S7(T)=E9
8820 PRINT"WHAT IF WE GET A FREEKICK UPFIELD BOSS?"
8822 PRINT " "
8824 FOR F1 =1 TO 3
8826 GOSUB 5480
8828 PRINTF1:" ":C$
8830 NEXT F1
8832 PRINT " "
8834 GOSUB 9960
8836 PRINT " "
8838 T5(T)=E9
8840 PRINT"HOW DO WE DEFEND AGAINST A FREEKICK?"
8842 PRINT " "
8844 FOR F1 =1 TO 3
8846 GOSUB 5680
8848 PRINTF1:" ":C$
8850 NEXT F1
8852 PRINT " "
8854 GOSUB 9960
8856 PRINT " "
8858 T6(T)=E9
8860 PRINT"WHAT SHALL WE DO
WITH OUR CORNERS?"
8862 PRINT " "
8864 FOR F1 =1 TO 3
8866 GOSUB 5530
8868 PRINTF1:" ":C$
8870 NEXT F1
8872 PRINT " "
8874 GOSUB 9960
8876 PRINT " "
8878 T7(T)=E9

```



```

8879 PRINT "WHAT ABOUT THEIR CORNERS?"
8880 PRINT " "
8881 FOR F1 = 1 TO 3:GOSUB 5730
8882 PRINT F1: " " : C$
8883 NEXT F1
8884 PRINT " "
8885 GOSUB 9960
8886 PRINT " "
8888 T8(T)=E9
8889 RETURN
8900 ON I GOTO 8910,8920,8930,8940,8950,8960
8905 RETURN
8910 IF J=1 THEN G2(1,1)=T9(J)
8915 IF J=2 THEN G2(3,1)=T9(J)
8918 GOTO 8905
8920 IF J=1 THEN G2(2,1)=T9(J)
8925 IF J=2 THEN G2(4,1)=T9(J)
8928 GOTO 8905
8930 IF J=1 THEN G2(1,2)=T9(J)
8935 IF J=2 THEN G2(3,2)=T9(J)
8938 GOTO 8905
8940 IF J=1 THEN G2(3,3)=T9(J)
8945 IF J=2 THEN G2(1,3)=T9(J)
8948 GOTO 8905
8950 IF J=1 THEN G2(4,2)=T9(J)
8955 IF J=2 THEN G2(2,2)=T9(J)
8958 GOTO 8905
8960 IF J=1 THEN G2(2,3)=T9(J)
8965 IF J=2 THEN G2(4,3)=T9(J)
8968 GOTO 8905
9000 REM PLAY GAME
9001 PRINT "G":PRINT A$(T): " U. " : A$(Z):PRINT " "
9002 B=0: C=0: B4=0: B5=0: B6=0
9003 GOSUB 3450
9004 GOSUB 8670
9005 GOSUB 3800
9010 Z3=0: Z4=0: W1=0
9012 K4=T4(S7(T), S7(Z))+30
9015 W3=R(T): W4=R(Z)
9016 W3=W3+2*H1(F(T), F(Z)): W4=W4+2*H1(F(Z), F(T))
9017 FOR I=1 TO 11: U2(I)=0: NEXT I
9018 U1=0
9019 PRINT "G":PRINT "THE REF BLOWS HIS WHISTLE TO START GAME"
9020 FOR I=1 TO K4
9025 Z1=T
9030 IF FNA(W3+W4)>R(T) THEN Z1=Z
9032 IF Z1=T THEN Z8=Z
9034 IF Z1=Z THEN Z8=T
9040 GOSUB 9800
9050 Z8$=" HAVE "
9052 GOSUB 9700
9055 PRINT A$(Z1): Z8$: "POSSESSION"
9056 PRINT " "
9070 IF Z1=T THEN GOSUB 9760
9080 IF Z1=Z THEN GOSUB 9780
9090 J$=Z5$+T4$(G1)+Z6$: GOSUB 15000
9092 PRINT " "
9094 GOSUB 9930
9100 Z4=Z2: Z3=Z1
9102 IF Z2=1 THEN U1=U1+1: U2(A3)=U2(A3)+1
9110 IF Z2=2 THEN GOSUB 9300
9112 IF I=22 THEN GOSUB 18000
9200 NEXT I
9201 H=T: A=Z
9202 PRINT "NOTHING ELSE OF SIGNIFICANCE HAPPENS DURING THE GAME"
9203 PRINT " ":PRINT "FINAL SCORE"
9204 PRINT " ":PRINT A$(T): B: " " : A$(Z): C
9205 P2=B: P3=C
9206 GOSUB 9900
9207 IF X3=0 THEN GOSUB 7100: RETURN
9208 IF B=C THEN 9211
9209 IF U=0 THEN RETURN
9210 GOSUB 7100: RETURN
9211 IF X5=1 THEN 9220
9212 Z3=0: Z4=0: K4=15
9214 X5=1
9216 PRINT "G":PRINT "EXTRA TIME BEING PLAYED"
9218 GOTO 9020
9220 X5=0
9222 PRINT "G":PRINT "PENALTIES WILL DECIDE THE RESULT"
9223 J$=" SHOTS WIDE"
9224 IF RND(1) >.5 THEN B=B+1: J$=" SCORES FROM THE SPOT"
9226 J$=K$(1+FNA(10))+J$: GOSUB 15000
9228 J$=" SHOTS WIDE"
9230 IF RND(1) >.5 THEN C=C+1: J$=" SCORES FROM THE SPOT"
9232 J$=Z9$(1+FNA(10))+J$: GOSUB 15000
9234 IF B=C THEN 9223
9236 PRINT "FINAL SCORE "
9238 PRINT A$(T): B: " " : A$(Z): C
9240 P2=B: P3=C
9245 GOSUB 9900
9250 RETURN
9300 REM SCORING CHANCE
9302 Z4=1
9305 G3=FNA(7)
9307 IF G1<8 THEN Z1$=Z6$
9309 IF G1>7 THEN Z1$=Z5$
9310 ON G3 GOSUB 9350,9350,9350,9350,9450,9450,9500
9349 RETURN
9350 REM NORMAL ATTACK
9355 G3=FNA(100)
9360 IF T3(S6(Z3), S5(Z8))>100-G3+B4 THEN 9400
9365 GOSUB 9375
9366 RETURN

```

```

9375 G3=FNA(9)
9380 Z1$=Z1$+A1$(G3)
9382 IF Z8=T THEN H$=" " : K$(1+FNA(4))
9384 IF Z8=Z THEN H$=" " : Z9$(1+FNA(4))
9385 IF G3<4 THEN Z1$=Z1$+H$
9386 IF G3<4 THEN J$=Z1$: GOSUB 15000: Z2=1: RETURN
9387 IF Z8=T THEN H$=" " : K$(1)
9388 IF Z8=Z THEN H$=" " : Z9$(1)
9389 IF G3<6 THEN J$=Z1$+H$: GOSUB 15000: RETURN
9390 J$=Z1$: GOSUB 15000: RETURN
9400 REM GOAL IS SCORED
9402 PRINT " "
9404 B4=B4+2.5
9405 J$=Z1$+A2$(FNA(9)): GOSUB 15000
9406 GOSUB 9410: GOSUB 9640
9407 IF I=K4 THEN RETURN
9408 PRINT "FROM THE KICK OFF"
9409 RETURN
9410 J$="IT'S A GOAL! " : A$(Z1)+ " HAVE SCORED": GOSUB 15000
9415 IF Z1=T THEN B=B+1
9420 IF Z1=Z THEN C=C+1
9422 D2=I+2
9423 IF RND(1)>.5 THEN D2=D2-1
9425 PRINT " "
9426 Z$=" MINUTES"
9427 IF D2=1 THEN Z$=" MINUTE"
9428 PRINT "SCORE AFTER " : D2: Z$
9430 PRINT " "
9432 PRINT A$(T): B: " " : A$(Z): C
9440 Z2=1
9449 RETURN
9450 REM CORNER
9452 PRINT " ":PRINT A$(Z1): " WIN A CORNER"
9455 PRINT " "
9456 IF Z1=T THEN Z1$=K$(5+FNA(6))
9458 IF Z1=Z THEN Z1$=Z9$(5+FNA(6))
9460 J$="FROM THE KICK THE BALL REACHES"+ " " : Z1$
: GOSUB 15000
9465 G3=FNA(100)
9470 IF T2(S6(Z3), S5(Z8))>100-G3+B6 THEN B6=B6+2: GOTO 9400
9475 GOSUB 9375
9499 RETURN
9500 REM SET-PIECE
9502 PRINT " "
9510 IF Z1=T THEN Z5$=Z9$(1+FNA(4))
9512 IF Z1=Z THEN Z5$=K$(1+FNA(4))
9514 J$=Z1$+ " IS BROUGHT DOWN BY " : Z5$+ " . THE REF AWARDS " : A$(Z1)
9516 J$=J$+ " A FREEKICK JUST OUTSIDE THE BOX": GOSUB 15000
9520 PRINT " "
9525 PRINT "FROM THE FREEKICK "
9530 G3=FNA(100)
9532 IF T3(S6(Z3), S5(Z8))>100-G3+B5 THEN B5=B5+2: GOTO 9400
9534 GOSUB 9375
9536 RETURN
9600 REM DETERMINE PLAYER TO PRINT
9640 REM BRING ON SUB?
9645 IF W1=1 THEN GOSUB 9900: RETURN
9647 PRINT "PRESS S TO BRING ON SUBSTITUTE"
9648 PRINT " ANY OTHER KEY TO CONTINUE"
9649 GET Z$
9650 IF Z$="" THEN 9649
9651 IF ASC(Z$)<>83 THEN RETURN
9655 PRINT "G"
9656 W1=1
9660 FOR J=2 TO 11:PRINT J: " " : K$(J): NEXT J
9665 PRINT "PLEASE TYPE A NUMBER FROM 2 TO 11"
9666 PRINT "TO INDICATE THE PLAYER TO BE SUBSTITUTED"
9667 INPUT K
9668 IF K>11 THEN 9667
9669 IF K<2 THEN 9667
9670 PRINT " ": J$=K$(K)+ " TRUDGES OFF TO BE REPLACED BY "
: K$(12): GOSUB 15000
9675 Z$=K$(K): K$(K)=K$(12): K$(12)=Z$
9680 W3=W3+(-(U1/10)+U2(K))*2
9685 RETURN
9700 REM DETERMINE IF POSS. REGAINED
9710 REM IF Z1<>1 THEN RETURN
9720 IF Z4<>1 THEN RETURN
9730 IF Z3<>Z1 THEN RETURN
9740 Z8$=" REGAIN "
9750 RETURN
9760 REM GENERATE 2 RANDOM PLAYERS
9762 A3=FNA(10)+1
9764 Z5$=K$(A3)
9766 IF G1>10 THEN Z6$=Z9$(FNA(10)+1)
9768 IF G1>10 THEN RETURN
9770 A4=FNA(10)+1
9772 IF A4=A3 THEN 9770
9774 Z6$=K$(A4)
9776 RETURN
9780 REM GENERATE 2 RANDOM PLAYERS
9782 A3=FNA(10)+1
9784 Z5$=Z9$(A3)
9786 IF G1>10 THEN Z6$=K$(FNA(10)+1)
9788 IF G1>10 THEN RETURN
9790 A4=FNA(10)+1
9792 IF A4=A3 THEN 9790
9794 Z6$=Z9$(A4)
9796 RETURN
9800 REM DETERMINE SUCCESS OF POSS.
9810 G1=FNA(20)
9820 IF G1<4 THEN Z2=1
9830 IF G1<4 THEN RETURN
9840 IF G1>17 THEN Z2=1

```

```

9850 IF G1>17 THEN RETURN
9860 Z2=2
9870 RETURN
9900 PRINT "PRESS ANY KEY TO CONTINUE"
9905 GET Z#
9910 IF Z#="" THEN 9905
9920 RETURN
9930 FOR U=1 TO 1200
9940 NEXT U
9950 RETURN
9960 REM INPUT NUMBER FROM 1 TO 3
9964 INPUT "ENTER A NUMBER BETWEEN 1 AND 3:";E#
9966 IF LEN(E#)<>1 THEN 9990
9967 IF ASC(E#)<49 THEN 9990
9968 IF ASC(E#)>51 THEN 9990
9969 E9=VAL(E#)
9980 RETURN
9990 PRINT "PLEASE RE-ENTER"
9992 GOTO 9964
11062 FOR I=1 TO 24
11064 F(I)=INT(RND(1)*3)+1
11066 NEXT I
11068 RETURN
11070 FOR I=1 TO 24
11080 READ A$(I)
11090 NEXT I
11092 RETURN
11094 FOR I=1 TO 3
11095 FOR J=1 TO 3
11096 READ H1(I,J):READ A1(I,J)
11097 NEXT J
11098 NEXT I
11100 RETURN
11110 FOR I=1 TO 20
11120 READ T4$(I)
11130 NEXT I
11140 RETURN
12100 REM GENERATE TEAM'S RATINGS
12110 FOR I=1 TO 3

```

Longer listings are best! That's the message you keep drumming into us on the telephone and by letter. Well, they don't come much longer than World Cup Manager and we hope you take the trouble to key it in. But for any Sharp owners who feel it will be too risky or time consuming to type it, a cassette of the game is available from Sharpsoft. It costs £5.75 and can be bought from Sharpsoft, 86/90 Paul Street, London EC1.

```

12605 Z9$(1)=" THEIR GOALKEEPER "
12610 RETURN
12615 REM GENERATE GOALSCORING MESSAGES
12620 FOR I=1 TO 9
12630 READ A1$(I)
12635 NEXT I
12640 FOR I=1 TO 9
12650 READ A2$(I)
12660 NEXT I
12670 RETURN
14000 REM CHECK VALIDITY OF NAME
14005 K3=46
14010 PRINT D$:INPUT K$(I)
14020 IF LEN(K$(I))>35 THEN PRINT "NAME IS TOO LONG PLEASE REINPUT"
14030 IF LEN(K$(I))>35 THEN 14010
14040 FOR K1=1 TO LEN(K$(I))
14050 K2=ASC(MID$(K$(I),K1,1))
14060 IF K2>90 THEN PRINT "ONLY LETTERS ARE ALLOWED-PLEASE RE-ENTER"
14070 IF K2>90 THEN GOTO 14010
14080 IF K2=32 THEN GOTO 14110
14082 IF K2+K3-90<3 THEN PRINT "WHAT SORT OF NAME IS THAT? PLEASE RE-ENTER"
14083 IF K2+K3-90<3 THEN 14010
14085 IF K2=39 THEN 14110
14088 IF K2=45 THEN 14110
14089 IF K2=46 THEN 14110
14090 IF K2<65 THEN PRINT "ONLY LETTERS ARE ALLOWED PLEASE RE-ENTER"
14100 IF K2<65 THEN GOTO 14010
14110 K3=K2
14115 NEXT K1
14120 RETURN
15000 IF LEN(J$)<=39 THEN PRINT J$
15005 IF LEN(J$)<39 THEN PRINT " "
15010 IF LEN(J$)<=39 THEN RETURN
15020 FOR J=39 TO 1 STEP -1
15030 IF ASC(MID$(J$,J,1))=32 THEN GOTO 16000
15040 NEXT J
15050 PRINT J$
15060 RETURN
16000 PRINT LEFT$(J$,J)
16010 J$=RIGHT$(J$,LEN(J$)-J)
16020 GOTO 15000
18000 PRINT " "
18010 PRINT "HALFTIME SCORE:"
18020 PRINT A$(T):B: " " :A$(Z):C
18030 PRINT "DO YOU WISH TO CHANGE YOUR TACTICS?"
18040 INPUT Z#
18045 IF ASC(LEFT$(Z$,1))=89 THEN GOSUB 8670
18050 GOSUB 9640
18060 RETURN

```

```

12120 R(I)=INT(RND(1)*10)+21
12130 NEXT I
12140 FOR I=4 TO 11
12150 R(I)=INT(RND(1)*10)+16
12160 NEXT I
12170 FOR I=12 TO 14
12180 R(I)=20
12190 NEXT I
12200 FOR I=15 TO 17
12210 R(I)=INT(RND(1)*5)+16
12220 NEXT I
12230 FOR I=18 TO 23
12240 R(I)=INT(RND(1)*5)+11
12250 NEXT I
12260 R(24)=INT(RND(1)*22)+11
12270 RETURN
12300 REM MATRIX OF MIDFIELD POSS.
12310 FOR I=1 TO 3
12320 FOR J=1 TO 3
12330 T4(I,J)=10
12340 NEXT J
12350 NEXT I
12360 T4(1,1)=5
12370 T4(3,3)=15
12380 REM MATRIX OF FREEKICKS TACTICS
12390 FOR I=1 TO 3
12400 FOR J=1 TO 3
12410 READ T1(I,J)
12420 NEXT J
12430 NEXT I
12440 REM MATRIX OF CORNER TACTICS
12450 FOR I=1 TO 3
12460 FOR J=1 TO 3
12470 READ T2(I,J)
12480 NEXT J
12490 NEXT I
12500 REM MATRIX OF ATTACK TACTICS
12510 FOR I=1 TO 3
12520 FOR J=1 TO 3
12530 READ T3(I,J)
12550 NEXT J
12560 NEXT I
12570 REM GENERATE OPPOSITION DEFAULT NAMES
12580 FOR I=1 TO 12
12590 Z9$(I)=" NO. "+STR$(I)
12600 NEXT I

```

